

Kapitel 13

Exit-Programme

Schnittstelle zwischen i-effect® und unternehmensspezifischem Workflow

i-effect® ermöglicht an verschiedenen, definierten Stellen der Verarbeitung den Aufruf von benutzerdefinierbaren Exit-Programmen. Zudem können zuvor definierte oder aus den Quelldateien extrahierte Daten an diese Exit-Programme übergeben werden.

Auf diese Weise ist eine optimale Einbindung aller i-effect®-Prozesse in die individuellen Systemabläufe eines Unternehmens möglich.

Bei den folgenden Beispielen für Exit-Programme steht jeweils ein konkretes Szenario im Hintergrund. Die Exit-Programme lösen jeweils die firmenspezifische Seite, während i-effect® als Standardanwendung im Zusammenspiel mit dem jeweiligen Exit-Programm den gewünschten Workflow im Power Systems-System realisiert.

Beispiel für ein Exit-Programm vom Typ 1 (RPGLE)

Das Beispielprogramm zeigt, wie ein Exit-Programm in **CVTSPL** benutzt werden kann, um den durch diesen Befehl erzeugten Datenstrom-Dateien sinnvollere Namen zu geben. Es wird in CVTSPL mit folgender Syntax angegeben:

```
EXITPGM(*LIBL/EXITREN *TYPE1)
```

Es wird unterstellt, dass die Option **EXITPGMPRM(*POS)** oder **EXITPGMPRM(*KEY)** verwendet wurde, um aus dem Bericht eine Information zu extrahieren, die dem Exit-Programm an neunter Stelle im Parameter **EXITPGMPRM** übergeben wird. Bei dieser Information kann es sich z. B. um Kundennamen, Bestellnummer, Rechnungsnummer o. a. handeln.

```
EXITPGMPRM(*POS)  
EXITPGMPOS((*ALL 10 60 10 *ROWCOL))
```

Die Datenstrom-Datei wird umbenannt. Aus dem durch CVTSPL gegebenen Namen wird der Name nnnnnn.pdf, wobei nnnnnn die aus dem Bericht extrahierte Nummer ist, welche als Parameter übergeben wurde. Die Quelldatei des folgenden Beispielprogramms finden Sie auch in der i-effect®-Arbeitsbibliothek in der **QRPGLESRC**-Datei.

```
H COPYRIGHT('© menten GmbH, 2003`)
**-----**
**                               tt                               **
**      mmmmm   eeeee   nnnnn   tt   eeeee   nnnnn   **
**      m m m e   e n   n tttt e   e n   n   **
**      m m m eeeeeee n   n tt   eeeeeee n   n   **
**      m m m e       n   n tt   e       n   n   **
**      m m m eeeeeee n   n ttt   eeeeeee n   n   **
**-----**
** E D V - B e r a t u n g   G m b H                               **
**-----**
**                               **
**      Systemberatung                                         **
**      IBM System AS/400                                       **
**      Individualprogrammierung                                 **
**      menten GmbH                                           **
**      Hauptstrasse 136 - 140                                  **
**      51465 Bergisch Gladbach                                **
**      Telefon +49 2202 2399 0                                 **
**      Telefax +49 2202-2399 23                               **
**      Email   info@menten-gmbh.de                           **
**      Homepage http://www.menten-gmbh.de                     **
**-----**
**                               **
** PROGRAMM      : EXITREN                                     **
**-----**
** BESCHREIBUNG : Sample 'TYPE 1' CVTSPL exit program for rename **
**      streamfile -> CustomerNr.pdf                           **
**-----**
** DATUM        : Aug. 2002                                    **
**-----**
** AUTOR        : Ilja Kolsow                                  **
**-----**
** Änderungsdatum :                                           **
** Änderungsgrund :                                           **
** Autor          :                                           **
**-----**
** PROTOTYPE FOR STANDARD EXIT PROGRAM PARAMETERS             **
**-----**
D Parameters      PR                               ExtPgm('EXITREN')
* Spooled file name
D   iaSplFile          10A
* Name of the job which created the spooled file
D   iaSplJob           10A
* User id of the job which created the spooled file
D   iaSplUser         10A
```

```
* Job number of the job which created the spooled file
D   iaSplJobNo        6A
* Spooled file number of the spooled file
D   iiSplNbr         10I 0
* Spooled file user data
D   iaUsrDta         10A
* Name of the stream file created
D   iaStmFile        128A
* IFS directory in which stream file was created
D   iaDir            256A
* Exit program parameter string extracted from spooled file
D   iaExitParm       1024A
**-----**
** PROTOTYPE FOR QCMDEXC                                     **
**-----**
**
D Command          PR                               ExtPgm('QCMDEXC')
D   iaCommand      32767A   CONST OPTIONS(*VARSIZE)
D   inCmdLen       15P 5   CONST
**-----**
** STANDARD EXIT PROGRAM PARAMETERS                         **
**-----**
D Parameters      PI
* Spooled file name
D   iaSplFile          10A
* Name of the job which created the spooled file
D   iaSplJob           10A
* User id of the job which created the spooled file
D   iaSplUser         10A
* Job number of the job which created the spooled file
D   iaSplJobNo        6A
* Spooled file number of the spooled file
D   iiSplNbr         10I 0
* Spooled file user data
D   iaUsrDta         10A
* Name of the stream file created
D   iaStmFile        128A
* IFS directory in which stream file was created
D   iaDir            256A
* User-definable exit program parameter.
D   iaExitParm       1024A
**
D laOldName        S           1024A
D laNewName        S           1024A
**
D QUOTE            C           ''''
**
D i                S           10I 0
**-----**
```

```

** main procedure **
**-----**
**
* If customer number found and passed as parameter, rename stream file
C          IF          iaExitParm <> *BLANKS
**
* Set up full path name to stream file
C          IF          iaDir = *BLANKS
C          EVAL          laOldName = QUOTE
C                          + %trim(iaStmFile)
C                          + QUOTE
C          ELSE
C          EVAL          laOldName = QUOTE
C                          + %trim(iaDir)
C                          + '/'
C                          + %trim(iaStmFile)
C                          + QUOTE
C          ENDIF
**
* Set up new path name for file using customer number in parameter
C          EVAL          laNewName = QUOTE
C                          + %trim(%subst(iaExitParm:1:10))
C                          + '.pdf'
C                          + QUOTE
**
* Rename the stream file
C          CALLP(E) Command( 'REN '
C                          + laOldName
C                          + ' '
C                          + laNewName:2053)
**
C          dow          %error
**
C          EVAL          i = i + 1
* Set up new path name for file using customer number in parameter
C          EVAL          laNewName = QUOTE
C                          + %trim(%subst(iaExitParm:1:10))
C                          + '_' + %trim(%editc(i:'Z'))
C                          + '.pdf'
C                          + QUOTE
* Rename the stream file
C          CALLP(E) Command( 'REN '
C                          + laOldName
C                          + ' '
C                          + laNewName:2053)
C          EndDo
C          ENDIF
**
C          RETURN

```

Beispiel für ein Exit-Programm vom Typ 1 (CLLE)

Dieses Beispielprogramm zeigt, wie ein Exit-Programm in **CVTSPL** benutzt werden kann, um den durch diesen Befehl erzeugten Datenstrom-Dateien sinnvollere Namen zu geben. Die Quelldatei des folgenden Beispielprogramms finden Sie auch in der i-effect®-Arbeitsbibliothek in der **QCLESRC**-Datei.

```

/*-----*/
/*          tt          */
/*          mmmmmm   eeeee   nnnnnn   tt   eeeee   nnnnnn   */
/*          m m m e   e n   n tttt e   e n   n          */
/*          m m m eeeeeee n   n tt   eeeeeee n   n          */
/*          m m m e     n   n tt   e     n   n          */
/*          m m m eeeeeee n   n ttt   eeeeeee n   n          */
/*          */
/* E D V - B e r a t u n g   G m b H          */
/*          */
/*          Systemberatung          */
/*          IBM System AS/400          */
/*          Individualprogrammierung          */
/*          */
/*          menten GmbH          */
/*          Hauptstrasse 136 - 140          */
/*          51465 Bergisch Gladbach          */
/*          Telefon +49 2202 2399 0          */
/*          Telefax +49 2202-2399 23          */
/*          Email   info@menten-gmbh.de          */
/*          Homepage http://www.menten-gmbh.de          */
/*-----*/
/*          PROGRAMM      : GZPEXIT          */
/*          */
/*          BESCHREIBUNG : Sample 'TYPE 1' CVTSPL exit program for rename          */
/*                          streamfile -> CustomerNr.pdf          */
/*          */
/*          DATUM        : September 2002          */
/*          */
/*          AUTOR        : Ilja Kolosow          */
/*          */
/*-----*/
/*          Änderungsdatum :          */
/*          Änderungsgrund :          */
/*          Autor          :          */
/*-----*/
PGM          PARM(&FILENAME &JOBNAME &USERID &JOBNBR +
&FILENBR &USERDTA &PDFNAME &PDFPATH +
&USRPARM)
COPYRIGHT TEXT('© menten GmbH, 2003')

```

```

/* Spooled file name */
      DCL      VAR(&FILENAME) TYPE(*CHAR) LEN(10)
/* Name of the job which created the spooled file */
      DCL      VAR(&JOBNAME) TYPE(*CHAR) LEN(10)
/* User id of the job which created the spooled file */
      DCL      VAR(&USERID) TYPE(*CHAR) LEN(10)
/* Job number of the job which created the spooled file */
      DCL      VAR(&JOBNBR) TYPE(*CHAR) LEN(10)
/* Spooled file number of the spooled file */
      DCL      VAR(&FILENBR) TYPE(*DEC) LEN(4 0)
/* Spooled file user data */
      DCL      VAR(&USERDTA) TYPE(*CHAR) LEN(10)
/* Name of the stream file created */
      DCL      VAR(&PDFNAME) TYPE(*CHAR) LEN(128)
/* IFS directory in which stream file was created */
      DCL      VAR(&PDFPATH) TYPE(*CHAR) LEN(128)
/* Exit program parameter string extracted from spooled file */
      DCL      VAR(&USRPARM) TYPE(*CHAR) LEN(1024)
      DCL      VAR(&FILE) TYPE(*CHAR) LEN(512)
      DCL      VAR(&FILE1) TYPE(*CHAR) LEN(512)
      DCL      VAR(&PDFNAME1) TYPE(*CHAR) LEN(128)

/**/
/* Globale Fehlerbehandlung */
/**/
      MONMSG      MSGID(RPG0000 CPF0000 GZI0000 CEE0000) +
                EXEC(GOTO CMDLBL(ERROR))

/**/
/* Name der soeben erzeugten PDF-Datei */
/**/
      CHGVAR      VAR(&FILE) VALUE(&PDFPATH *TCAT '/' *TCAT +
                &PDFNAME)

/**/
/* An dieser Stelle müssen die Werte der nachfolgenden Variablen */
/* gefüllt werden. */
      CHGVAR      VAR(&PDFNAME1) VALUE(&USRPARM)
      CHGVAR      VAR(&PDFNAME1) VALUE(&PDFNAME1 *TCAT '.pdf')
      REN         OBJ(&FILE) NEWOBJ(&PDFNAME1)

      RETURN

/**/
/* Fehlerbehandlung */
/**/
ERROR:
      SNDPGMMSG  MSGID(CPF9898) MSGF(*LIBL/QCPFMSG) +
                MSGDTA('Das i-effect Exit-Programm wurde +
                abnormal beendet') MSGTYPE(*ESCAPE)

END:      ENDPGM

```

Beispiel für ein Exit-Programm vom Typ 2 (RPGLE)

Im Beispielprogramm Typ 2 verwendet das Exit-Programm **EXITSNDEML** eine Parameter-Liste vom **Typ 2 (*TYPE2)**.

Es soll aufgerufen werden für jede neu erstellte Datenstrom-Datei, um diese sogleich automatisch per eMail an den Kunden zu versenden, für den sie bestimmt ist.

EXITPGM(*LIBL/EXITSNDEML *TYPE2)

Es wird unterstellt, dass die **EXITPGMPRM(*POS)** oder **EXITPGMPRM(*KEY)** Option verwendet wurde, um aus dem Bericht die Kundennummer zu extrahieren. Mit dieser Kundennummer wird gewöhnlich in der Kundendatenbank die Adresse des Kunden ermittelt. Der i-effect®-**SNDEMAIL**-Befehl wird dann verwendet, um die Datenstrom-Datei an den entsprechenden Kunden zu versenden. Die Quelldatei des folgenden Beispielprogramms wie auch die **CUSTFILE** Quelldatei finden Sie auch in der i-effect®-Arbeitsbibliothek in den **ORPGLESRC** und **QDDSSRC**-Dateien.

```

H COPYRIGHT('© menten GmbH, 2003')
**-----**
**                                     tt                                     **
**      mmmmmmm   eeeee   nnnnnn   tt   eeeee   nnnnnn   **
**      m m m e   e n   n   tttt e   e n   n   **
**      m m m eeeeeee n   n   tt   eeeeeee n   n   **
**      m m m e   n   n   tt   e   n   n   **
**      m m m eeeee n   n   ttt   eeeee n   n   **
**
** E D V - B e r a t u n g   G m b H
**
**                               Systemberatung
**                               IBM System AS/400
**                               Individualprogrammierung
**
**                               menten GmbH
**                               Hauptstrasse 136 - 140
**                               51465 Bergisch Gladbach
**                               Telefon +49 2202 2399 0
**                               Telefax +49 2202-2399 23
**                               Email   info@menten-gmbh.de
**                               Homepage http://www.menten-gmbh.de
**-----**
** PROGRAM NAME: EXITSNDEML
**
** APPLICATION : CVTSPL command
**
** LANGUAGE    : ILE RPG
**
** AUTHOR      : Ilja Kolosow

```

```

**
** DATE WRITTEN: Sep. 2002
**
** DESCRIPTION : Sample 'TYPE 2' CVTSPL exit program for sending a
**               stream file via e-mail using the SNDEMAIL command
**               of i-effect
**-----**

* Customer file

FCustFile  IF  E           K DISK  RENAME(CUSTFILE:CUSTFILER)

**-----**
** PROTOTYPE FOR STANDARD EXIT PROGRAM PARAMETERS
**-----**

D Parameters      PR           EXTPGM('EXITSNDEML')

* Spooled file name

D  iaSplFile           10A

* Name of the job which created the spooled file

D  iaSplJob            10A

* User id of the job which created the spooled file

D  iaSplUser           10A
* Job number of the job which created the spooled file

D  iaSplJobNo          6A

* Spooled file number of the spooled file

D  iiSplNbr            10I 0

* Spooled file user data

D  iaUsrDta            10A

* Name of the stream file created

D  iaStmFile           128A

* IFS directory in which stream file was created

D  iaDir                256A

* Page number of the first page in the range processed

```

```

D  iiFirstPage         10I 0

* Page number of the last page in the range processed

D  iiLastPage          10I 0

* Exit program parameter count

D  iiParmCount         10I 0

* User-definable exit program parameter strings. Up to 150 of these
* may follow. Define as many as you need.

D  iaUserParam1        999A      OPTIONS(*NOPASS)
D                               VARYING

D  iaUserParam2        999A      OPTIONS(*NOPASS)
D                               VARYING

D  iaUserParam3        999A      OPTIONS(*NOPASS)
D                               VARYING

**-----**
** PROTOTYPE FOR QCMDEXC
**-----**

D Command            PR           ExtPgm('QCMDEXC')
D  iaCommand          32767A      CONST OPTIONS(*VARSIZE)
D  inCmdLen           15P 5 CONST

**-----**
** STANDARD EXIT PROGRAM PARAMETERS
**-----**

D Parameters      PI

D  iaSplFile           10A
D  iaSplJob            10A
D  iaSplUser           10A
D  iaSplJobNo          6A
D  iiSplNbr            10I 0
D  iaUsrDta            10A
D  iaStmFile           128A
D  iaDir                256A
D  iiFirstPage         10I 0
D  iiLastPage          10I 0
D  iiParmCount         10I 0
D  iaUserParam1        999A      OPTIONS(*NOPASS)
D                               VARYING
D  iaUserParam2        999A      OPTIONS(*NOPASS)

```

```

D          VARYING
D   iaUserParam3      999A      OPTIONS(*NOPASS)
D          VARYING

D waCustomerNo   S      10A
D waCommand      S      1024A
D wnCmdLen       S      15P 5

D QUOTE         C          ''''

* This example assumes that the EXITPGMPRM, EXITPGMPOS and EXITPGMKEY
* parameters have been used to extract the customer number from the
* spooled file, and that the first user-definable parameter contains the
* customer number (assumed to be 10 characters for the purposes of
* this example).

* Make sure at least one parameter was passed and that the parameter
* passed is not too short

C          IF          iiParmCount < 1
C                  or %len(iaUserParam1) < %size(waCustomerNo)
* Insert appropriate error handling here
C          RETURN
C          ENDIF

* Extract the customer number from the parameter string

C          EVAL      waCustomerNo = %subst(iaUserParam1      :
C                  1      :
C                  %len(waCustomerNo))

* Retrieve the customer's e-mail address from the customer file

C   waCustomerNo CHAIN   CustFile
C          IF          not %found(CustFile)
* Insert appropriate error handling here
C          RETURN
C          ENDIF

* Build the SNDEMAIL command
C          EVAL      waCommand = 'SNDEMAIL '
C                  + 'RECIPIENT(' + QUOTE
* „CustEMail“ is assumed to be the customer's email address field
* from the customer file
C                  + %trim(CustEMail) + QUOTE
C                  + ') '

* DSTD parameter below is the „subject“ line that will appear for
* the e-mail. Substitute your choice of subject line
C                  + 'SUBJECT('

```

```

C          + QUOTE
C          + 'Your invoice from i-effect '
C          + QUOTE
C          + ') '
* MSG parameter below is a brief message to accompany the attachment.
* Substitute your choice of message.
C          + 'MSGTEXT('
C          + QUOTE
C          + 'Your invoice is'
C          + ' attached. Thank you for your '
C          + 'custom.'
C          + QUOTE
C          + ') '
* The document name and folder names are passed into this program from
* CVTSPL. The document name is generated from the OUTPFILE parameter
* with the addition of a numeric suffix for each spooled file created.
C          + 'ATTACH(' + QUOTE
C          + %trim(%trim(iaDir) + '/')
C          + %trim(iaStmFile)) + QUOTE
C          + ') '
C          eval      wnCmdLen = %len(waCommand)

* e-mail the stream file to the customer
C          CALLP(E)   Command(waCommand:wnCmdLen)

C          IF          %error
* Insert appropriate error handling here
C          ELSE
* Send OK message
C          ENDIF
C          RETURN

```

EXITSNDEML – Verwendungsbeispiel

In folgenden Beispiel wird die Spooldatei **INVNEW1** aus der **QSAMPLER Bibliothek (IBM AFP Beispiele)** verwendet.

Es wird unterstellt, dass die **EXITPGMPRM(*POS)** oder **EXITPGMPRM(*KEY)** Option verwendet wurde, um aus dem Bericht die Kundennummer zu extrahieren. Mit dieser Kundennummer wird gewöhnlich in der Kundendatenbank die eMail-Adresse des Kunden ermittelt. Der i-effect®-**SNDEMAIL**-Befehl wird dann verwendet, um die Datenstrom-Datei an den entsprechenden Kunden zu versenden. Die Quelldatei des folgenden Beispielprogramms wie auch die **CUSTFILE** Quelldatei finden Sie auch in der i-effect®-Arbeitsbibliothek in den **QRPGLESRC** und **QDDSSRC** -Dateien.

Mit allen Spooldateien arbeiten

Auswahl eingeben und Eingabetaste drücken.

- 1=Senden 2=Ändern 3=Anhalten 4=Löschen 5=Anzeigen 6=Freigeben
7=Nachrichten 8=Attribute 9=Mit Druckstatus arbeiten

Opt	Datei	Warte- schlange	Bibliothek	ASP	Zuletzt benutzt	Spool- Datei	Größe (K)
5	INVNEW1	CUSTOMER	KOLOSOW	1	24.08.04		44

Spooldatei anzeigen

```

Datei . . . . . : INVNEW1                               Seite/Zeile 1/18
Steuerung . . . . +1_____                           Spalten 1 - 78
Suchen . . . . . : _____
*...+...1...+...2...+...3...+...4...+...5...+...6...+...7...+...
      100          31300          13.02.03          3/13/03
      BEST WAY          13.02.03          NET 30          YOUR PRINTER RE
1 CT  00000300  HIGH ALTITUDE WATERMELON          1.01          1.01
1 PK  01100517  SPARTAN SEEDS                      2.39          2.39
9 PK  04569870  NORTHERN LITE BLUE SPRUCE          858.32          7,724.88
12 BX 11005004  BUSH GREEN SEEDS                          2.50          30.00
12 CT 11005011  LASSO RED SEEDS                          892.23          10,706.76
26 PK 11005018  EARLY BANTAM SEEDS                        .38           9.88
5 BX  11057893  AFRICAN DAISY, SEEDS                    2.35          11.75
1 PK  15975365  HEAVY OAK                                129.09          129.09
33 BX 32746510  HOPS BREWING LIGHT                       1.20          39.60
6 EA  46578913  SEED SURVEYING SITE                      50.00          300.00
2 BX  56413213  POT POT                                  7.65          15.30
80 PK 65412384  SEED SCRUBBER                            888.79          71,103.20
1 PK  84512023  OREGON SPRING TOMATO SEED                .97           .97
2 DZ  96325874  PINEAPPLE-ORANGE SEEDS                   1.29          2.58

      Weitere ...
    
```

Spooldatei(en) konvertieren (CVTSPL)

Auswahl eingeben und Eingabetaste drücken.

```

Aufteilen per Position:          SPLITPOS
Zeilen-Nr. oder y Koordinate    > 18_____
Zeichen-Pos. o. x-Koordinate    > 17_____
Länge . . . . .                 3_____
Messmethode . . . . .           *ROWCOL
Vergleich . . . . .             *NE
Zu vergleichende Zeichenkette   *PRV_____

Exit Programme:                 EXITPGM
Dateierstellung . . . . .       > EXITSNDEML
    
```

```

Bibliothek . . . . .           > IEFFECT___
Parameterformat . . . . .     > *TYPE2
Seitenumbruch . . . . .      *NONE___
Bibliothek . . . . .           _____
Parameterformat . . . . .     *TYPE1
Dateiende . . . . .           *NONE___
Bibliothek . . . . .           _____
Parameterformat . . . . .     *TYPE1
Exit-Programm vor Dateierstell *NONE___
Bibliothek . . . . .           _____
Parameterformat . . . . .     *TYPE1

Exit Programmparameter:        EXITPGMPRM
Parameter typ . . . . .        *POS___
CCSID der Parameter Daten . . . *SPLF___

Exit Progr.Param. String-Pos.: EXITPGMPOS _
Seitenzahl . . . . .          > 1_____
Zeilen-Nr. oder y Koordinate  > 18_____
Zeichen-Pos. o. x-Koordinate  > 17_____
Länge . . . . .               > 3_____
Messmethode . . . . .         *ROWCOL

      + für weitere Werte _
    
```

Beispiel Zuordnung Kundennummer -> eMail Adresse:

Physische Teildatei anzeigen

```

Datei . . . . . : CUSTFILE          Bibliothek . . . . : GZIPR7
Teildatei . . . : CUSTFILE1        Satz . . . . . : 1
Steuerung . . . : _____       Spalte . . . . . : 1
Suchen . . . . . : _____
*...+...1...+...2...+...3...+...4...+...5...+...6...+...7...+...
100      100.MUSTERMANN@MENTEN-GMBH.DE          TEST CUSTOMER SNDE
136      136.MUSTERMANN@MENTEN-GMBH.DE          TEST CUSTOMER SNDE
141      141.MUSTERMANN@MENTEN-GMBH.DE          TEST CUSTOMER SNDE
154      154.MUSTERMANN@MENTEN-GMBH.DE          TEST CUSTOMER SNDE

      ***** DATENENDE *****
    
```

CVTSPL FILE (INVNEW1)

```

JOBNAM(011085/KOLOSOW/QPRTJOB)
SPLNBR(105)
OUTPPFILE('%TMP%%SPLFILENAME%')
PAGESIZE(*A4 *PORTRAIT)
SPLIT(*POS)
SPLITPOS(18 17 3)
EXITPGM(IEFFECT/EXITSNDEML *TYPE2)
EXITPGMPRM(*POS)
EXITPGMPOS((1 18 17 3))

...
SPLIT(*POS) Verarbeitung für Parameter Element 1 wählte Text „100“.
EXITPGMPRM(*POS) Verarbeitung für Parameter Element 1 wählte Text „100“.
    
```

SPLIT(*POS) Verarbeitung für Parameter Element 1 wählte Text „136“.
EXITPGMPRM(*POS) Verarbeitung für Parameter Element 1 wählte Text „136“.

SPLIT(*POS) Verarbeitung für Parameter Element 1 wählte Text „ 56“.
EXITPGMPRM(*POS) Verarbeitung für Parameter Element 1 wählte Text „ 56“.

SPLIT(*POS) Verarbeitung für Parameter Element 1 wählte Text „141“.
EXITPGMPRM(*POS) Verarbeitung für Parameter Element 1 wählte Text „141“.

SPLIT(*POS) Verarbeitung für Parameter Element 1 wählte Text „154“.
EXITPGMPRM(*POS) Verarbeitung für Parameter Element 1 wählte Text „154“.
...


Spooldatei INVNEW1 konvertiert in Datenstrom-Datei /TMP/INVNEW1_30.pdf.
eMail wird in Archivdatei /gzip400/archive/mail/out/M2133.EML gesichert.
Verbindung zu Mailserver davad aufgebaut
An: 100.MUSTERMANN@MENTEN-GMBH.DE
Anhang /TMP/INVNEW1_30.pdf gesendet.
Verbindung zu Mailserver davad beendet
SNDEMAIL normal beendet.
Exit programm EXITSNDEML ended normally.

Spooldatei INVNEW1 konvertiert in Datenstrom-Datei /TMP/INVNEW1_31.pdf.
eMail wird in Archivdatei /gzip400/archive/mail/out/M2134.EML gesichert.
Verbindung zu Mailserver davad aufgebaut
An: 136.MUSTERMANN@MENTEN-GMBH.DE
Anhang /TMP/INVNEW1_31.pdf gesendet.
Verbindung zu Mailserver davad beendet
SNDEMAIL normal beendet.
Exit programm EXITSNDEML ended normally.

Spooldatei INVNEW1 konvertiert in Datenstrom-Datei /TMP/INVNEW1_32.pdf.
Exit programm error: customer number 56 not found in CUSTFILE!.
Spooldatei INVNEW1 konvertiert in Datenstrom-Datei /TMP/INVNEW1_33.pdf.
eMail wird in Archivdatei /gzip400/archive/mail/out/M2135.EML gesichert.
Verbindung zu Mailserver davad aufgebaut
An: 141.MUSTERMANN@MENTEN-GMBH.DE
Anhang /TMP/INVNEW1_33.pdf gesendet.
Verbindung zu Mailserver davad beendet
SNDEMAIL normal beendet.
Exit programm EXITSNDEML ended normally.


Spooldatei INVNEW1 konvertiert in Datenstrom-Datei /TMP/INVNEW1_34.pdf.
eMail wird in Archivdatei /gzip400/archive/mail/out/M2136.EML gesichert.
Verbindung zu Mailserver davad aufgebaut
An: 154.MUSTERMANN@MENTEN-GMBH.DE
Anhang /TMP/INVNEW1_34.pdf gesendet.
Verbindung zu Mailserver davad beendet
SNDEMAIL normal beendet.
Exit programm EXITSNDEML ended normally.

Spooldatei INVNEW1 in *PDF konvertiert.
Spooldatei INVNEW1 aus Job QPRTJOB/KOLOSOW/011085 Nummer 000105
erfolgreich verarbeitet.
0000000001 Spoolinträge wurden erfolgreich konvertiert.
i-effect Verarbeitung beendet.



Adobe Acrobat Professional - [INVNEW1_35.pdf]

400 CPU Parkway
Vegetation, NJ 55090



Office: 555-499-2367
Fax: 555-415-9794

44444444444444444444444444444444

IMPROVED PRINTING CORP
PERFORMANCE BOULEVARD
PRINTERSVILLE
CO 45789-2637

GAME

Customer Number: 100 Invoice Number: 31300 Invoice Date: 13.02.03 Payment Date: 3/13/03

Ship Via: BEST WAY Shipped Date: 13.02.03 Terms: NET 30 Salesman: YOUR PRINTER REP

Qty	UOM	Item #	Item Description	Price	Extension
1	CT	00000300	HIGH ALTITUDE WATERMELON	1.01	1.01
1	PK	01100517	SPARTAN SEEDS	2.39	2.39
9	PK	04569870	NORTHERN LITE BLUE SPRUCE	858.32	7,724.88
12	BX	11005004	BUSH GREEN SEEDS	2.50	30.00
12	CT	11005011	LASSO RED SEEDS	892.23	10,706.76
26	PK	11005018	EARLY BANTAM SEEDS	.38	9.88
5	BX	11057893	AFRICAN DAISY, SEEDS	2.35	11.75
1	PK	15975365	HEAVY OAK	129.09	129.09
33	BX	32746510	HOPS BREWING LIGHT	1.20	39.60
6	BA	46578913	SEED SURVEYING SITE	50.00	300.00
2	BX	56413213	POT POT	7.65	15.30
80	PK	65412384	SEED SCRUBBER	888.79	71,103.20
1	PK	84512023	OREGON SPRING TOMATO SEED	.97	.97
2	DS	96325874	PINEAPPLE-ORANGE SEEDS	1.29	2.58
11	BX	98412006	BLACK BEAUTY ZUCCHINI	2.30	25.30
5	BA	98546320	FROZEN JUICE PROCESSOR	109.90	549.50

Thank You
Because you have ordered
over \$500 of seeds this
year, on your next seed
order you will receive
a 10% discount.

This invoice overlay designed using IBM AFP Utilities/400

Total Due \$90,652.21


Return this tear-off strip with your payment.

Payment is due by: 3/13/03

Make Checks Payable to: Super Sun Seeds

Amount Due is: \$90,652.21

IMPROVED PRINTING CORP
PERFORMANCE BOULEVARD
PRINTERSVILLE
CO 45789-2637



00010009065221

210 x 297 mm

1 von 1